

10680-2016

## Key Features in ODS Graphics for Efficient Clinical Graphing

Yuxin (Ellen) Jiang, Biogen, Cambridge, MA

### ABSTRACT

High-quality effective graphs not only enhance understanding of the data but also facilitate regulators in the review and approval process. In recent SAS<sup>®</sup> releases, SAS has made significant progress toward more efficient graphing in ODS Statistical Graphics (SG) procedures and Graph Template Language (GTL). A variety of graphs can be quickly produced using convenient built-in options in SG procedures.

With graphical examples and comparison between SG procedures and traditional SAS/GRAPH procedure in reporting clinical trial data, this paper highlights several key features in ODS Graphics to efficiently produce sophisticated statistical graphs with more flexible and dynamic control of graphical presentation including 1) Better control of axes in different scales and intervals; 2) Flexible ways to control graph appearance; 3) Plots overlay in single-cell or multi-cell graphs; 4) Enhanced annotation; 5) Classification panel of multiple plots with individualized labeling.

### Introduction

Since SAS 9.2, SAS has brought out a set of improved capabilities in ODS Graphics to effectively create high quality graphics. Many commonly used clinical graphs can be more efficiently created or improved using those new components listed below. These procedures have been continually enhanced in SAS<sup>®</sup> 9.3 and 9.4 in Base SAS software.

- ODS Graphics editor: For non-programmer to edit and annotate ODS graphics in SGE files.
- ODS Graphics designer: Provides interactive graphical user interface for designing and customizing graphs easily and outputs GTL code behind the scene.
- Statistical Graphics (SG) procedures: utilize a set of predefined templates and built-in options to efficiently create single cell or multi-cell graphics, including:
  - SGPLOT: creates many types of single cell graphs with layout overlay
  - SGPANEL: creates multi-cell panels by one or more classification variables
  - SGSCATTER : creates multi-cell comparative and matrix plots.
- Graph Template Language (GTL): provides direct access to the underlying language for the default templates, to modify these templates or to create your own highly customized graphics.

Many graphs can be done quickly and efficiently using SG procedures, whereas sometimes they could take multiple steps or be impossible to make using traditional GPLOT in SAS/GRAPH. This paper is intended to share experience with other SAS users, especially GPLOT users who are


interested in trying out SG procedures. With the examples of comparison of figures using ODS SG procedures and traditional GPLOT procedure in SAS/GRAPH, several key features in ODS Graphics are demonstrated to efficiently produce highly customized graphs with better control of graph appearance in easily automated coding.

### Feature1: Better control of axes in different scales and intervals


SGPLOT has options for four types of axes (TYPE= DISCRETE | LINEAR | LOG | TIME). The default axis type is automatically chosen based on the variable types on the axis.

- In SGPLOT, if a numeric variable on the axis, a linear scale axis will be automatically set as default for the axis. As in Fig.1, it's easy to get unevenly spaced sample collection timepoints on axis using values=(0 1 3 5 7 10 14 21 28), and use a macro parameter to passing the string of tick mark values.
- In GPLOT, it's not straightforward when data are collected at uneven intervals, when use order =(0 1 3 5 7 10 14 21 28) at AXIS statement, a plot with equal-width intervals will be incorrectly generated and SAS log will have warning. This could bring challenges when trying to automate the code. Several papers had addressed this issue, and all of them had to inevitably involve ANNOTATE dataset or FORMAT to control the uneven tick marks on axis [Lin, 2007; Watts, 2009; ],

**Fig. 1 SGPLOT**



**Fig. 1: GPLOT**



### Feature 2: Provide flexible ways to control graph appearance


Since ODS graphics is a part of Output Delivery System (ODS), we can have several ways to control the graph appearance and the attributes of an graph element such as font, color, symbol and pattern, etc.

- ① Use or modify pre-defined ODS style templates to control graph appearance. In Fig. 2, we modified default listing style template to create MYSTYLE to customize the Marker Symbol . One style template can be used to standardize graph appearance for sorted GROUPs for same or different type of figures.

- ② STYLEATTRS, a new enhancement in SAS 9.4, provides convenient access to modify frequently changed figure attributes such as colors, line patterns and data symbols in SGPLOT and GTL .
- ③ Use in statement LINEATTRS or MARKERATTRS options to customized individual plot attributes eg. lineattrs=(color=black) and markerattrs=(color=black) will overwrite the previous Style setting to make the line and symbol color black.

In GPLOT, those attributes such as symbol, color and size have to be set by SYMBOL statement or using annotation datasets outside Proc GPLOT. The options like AXIS, PATTERN, SYMBOLS which are used in the GOPTION STATEMENT have been replaced using XAXIS, YAXIS, LINEATTRS=, MARKERATTRS=, and FILLATTRS= options within each SGPLOT statement.

Fig. 2



```
%sganno;
```

```
data anno;
```

```
%sgarrow(x1=14,y1=285,x2=14,y2=270,
  shape='filled', linethickness=1, scale=0.2,
  x1space='datavalue', x2space='datavalue',
  y1space='datavalue', y2space='datavalue' );
```

```
%sgtext(label="Midpoint", x1=13,y1=300,
  width=22, widthunit="data", textsize=10.5,
  textcolor='black',x1space='datavalue',
  y1space='datavalue', transparency=0.2,
  anchor='topleft' );
```

```
run;
```

The SAS code below generates the Fig. 2 above.

```
proc template; ①
  define style mystyle;
  parent=styles.listing;
  style GraphData1 from GraphData1 / MarkerSymbol = "circle";
  style GraphData2 from GraphData2 / MarkerSymbol = "X";
  end;
run;

ods rtf file = "&filename..rtf" style=mystyle ① nogtitle;
ods graphics / reset imagename="&filename" border=off ;


proc sgplot data=final sganno=anno;
  Format xaxis xaxis. trtpn trt. ;
  styleattrs datasymbols=(Circle X) datacontrastcolors=(purple green); [② SAS 9.4]
  series x=xaxis y=mean /group=trtpn lineattrs=(size=1) ③ name='l';
  scatter x=xaxis y=mean /group=trtpn yerrorlower=lower yerrorupper=upper
  markerattrs=(size=7) ③ name='s';
  xaxistable n / x=xaxis class=trtpn location=inside colorgroup=trtpn separator; ④[SAS 9.4]
  keylegend 'l' / border position=topright location=inside across=1;
  xaxis min=0 values=(0 1 3 5 7 10 14 21 28) valueattrs=(size=9) minor fitpolicy=rotate
  label='Days';
  yaxis min=&yAxisMin max=&yAxisMax values=(100 to 350 by 50) grid minor label="Results" ;
run;
ods rtf close;
```

### Feature 3: Plots overlay in single-cell or multi-cell graphs

Both SGPLOT procedure and GTL supports multiple plot statements that can be combined in single-cell or multi-cell by numerous ways to create effective graphs. Built on the layout overlay concept, each individual plot is in its own layout container. Same or different types of plots (with some limitation) can be overlaid together by SGPLOT or GTL. GTL is much more flexible when it comes to creating customized overlays with different types of figures in a dashboard look. As the example below, we overlaid a series plot, a barchart and a scatter plot in one graph. Note that scatter plot and regression line or series plot can be overlaid in one cell easily.

- ❶ PROC TEMPLATE and SGRENDER procedures are used to create and apply overall layout template.
- ❷ Two LAYOUT LATTICE statements define the position(rows=,columns=) and the size of each layout container (columnweights=)
- ❸ Three LAYOUT OVERLAY statements pairing up with END OVERLAY create blocks which define the plots in three layout container. The coding for each block can be generated from TMPLOUT option from SGPLOT or from ODS Graphics designer.

Fig. 3



The SAS code below generates the Fig. 3 above. The <option1> is provided in Appendix at the end of this paper.

```

proc template; ❶
define statgraph cd4cd8;
begingraph / datacontrastcolors=(darkblue darkred) datasymbols=(circle triangle)
            datalinepatterns=(dash solid);
entrytitle halign=center "Biomarker test summary" / textattrs=(size=11.7pt) haligncenter=graph;
  layout lattice /rows=1 columns=2 columnweights=(0.4 0.6) columngutter=.5cm; ❷
    layout lattice / rows=2 columns=1 rowgutter=.5cm; ❷
      layout overlay / xaxisopts=(label="Test A" labelAttrs=(Weight=bold)) ❸
                    yaxisopts=(griddisplay=on label="Test B" labelAttrs=(Weight=bold));
        scatterplot x=v1 y=v2/group=trtp1;
        regressionplot x=v1 y=v2/group=trtp1;
      endlayout;
      layout overlay / xaxisopts=(label="Day" labelAttrs=(Weight=bold)) ❸
                    yaxisopts=(griddisplay=on label="Mean of Test C");
        barchart category=day response=v3/group=grp stat=mean groupdisplay=cluster;
      endlayout;
    endlayout;

    layout overlay/<option1>; ❸
      seriesplot x=xvar y=mean_ratio / group=grpc display=(markers) name="1";
      scatterplot x=xvar y=mean_ratio/ group=grpc primary=true
                yerrorupper=upper_ratio yerrorlower=lower_ratio ;

      innermargin / align=bottom opaque=true separator=true;
        ❹ axistable value=n_ratio x=day/includemissingclass=false ) [SAS 9.4]
          headerLabel="Evaluable subjects (n)" labelPosition=min class=grpc
          valueattrs=(size=7) colorgroup=grpc;
        endinnermargin;

      discreteLegend "1" / Location=Inside across=1 halign=left valign=top Border=false;
    endlayout;
  endlayout;
endgraph;
end;
run;

```

## Feature 4: Enhanced annotation

As shown in Fig. 2 and Fig. 3, similar to the annotation facility ANNOTATE= option and %ANNOMAC in SAS/GRAPH, SAS has introduced SGANNO= option in the SGRENDER statement to specify the name of your annotation data set. %SGANNO macro function dictionary has also added for SGPLOT and GTL since SAS 9.4 to conveniently add annotation on figures such as arrow (%SGARROW), line (%SGLINE), and text (%SGTEXT) etc.

❹ AXISTABLE is a new feature in SAS 9.4, which creates an event plot of input data along an axis of an X-Y plot. With one succinct line of code, a table of counts(N) is printed on the bottom of the plot either inside (INNERMARGIN) or outside the plot. It saves lots of coding to create annotation datasets and lines up the numbers at bottom of the figure. The biggest benefit is that the AXISTABLE can adjust position accordingly to the scale and position of the plot which makes automation much easier.

## Feature 5: Classification panel of multiple plots with individualized labeling


The SGPPANEL procedure creates a panel of graph cells classified by one or more variables. This feature is very usefully to create by subject profiles. For example in Fig. 4 below,

- 1 Using PANELBY Subjid, we created a profile by subjects figure in a panel layout with one subject per cell.
- 2 Using DATALABEL, we added treatment group along each line and shown data value on day 1, made the graph more informative.

```
ods graphics / reset width=7in height=9in border=off;

proc sgpanel data=fix ;
  ❶ panelby id/ layout=panel rows=3 columns=2 spacing=10 ;
  series x=xvar y=yvar / group=grp lineattrs=(pattern=solid) markers ❷ datalabel=grpname
    name='l' ;
  scatter x=xvar y=yvar / group=grp markerattrs=(size=7) name='s';
  rowaxis min=1 max=100 minor type=log logbase=10 LOGSTYLE=LOGEXPAND valueattrs=(size=9)
    label= 'Results';
  colaxis min=0 values=(0 1 2 3 5 7) valueattrs=(size=9) fitpolicy=rotate label='Days'
    offsetmin=0.02;
  refline 5 10 50 / transparency = 0.8 ;
run;
```

Fig. 4



## Conclusion:

As SAS keep adding new enhancements in SGPLOT and GTL, it provides powerful flexibility to make clinical graph much more effective. Coding of graphs become much clearer and easier with ODS Graphics tools (SGPLOT, GTL). Author hope this paper could give you some insides for these new enhancements and make the metamorphosis from traditional SAS/GRAPH to the new frontier of ODS GRAPHICS easier.

## Acknowledgments:

I appreciate Biogen for giving me the opportunity to work on this paper and Mr. Mark McLaughlin and Mr. Arthur Collins for kindly reviewing the draft paper and providing valuable comments. All remaining errors are solely my responsibility.

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.

## Reference:

- Clinical Graphs using SAS 9.3, Sanjay Matange, SAS Institute Inc. SAS Global Forum 2013
- SAS® 9.4 ODS Graphics documents:  
<http://support.sas.com/documentation/cdl/en/grsggs/64979/PDF/default/grsggs.pdf>
- Creating Sophisticated Graphics using Graph Template Language. Kaitlyn McConville, Kristen Much, Rho, Inc., Chapel Hill, NC
- Generate a Customized Axis Scale with Uneven Intervals in SAS® - Automatically. Perry Watts, SAS® User, Elkins Park, PA 2009
- Tips and Tricks in Creating Graphs Using PROC GPLOT, Qin Lin, Bala Cynwyd, PA 2007

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Name: Yuxin (Ellen) Jiang

Enterprise: Biogen

Address: 300 Binney St, Suite 94A49

City, State ZIP: Cambridge, MA 02142

Work Phone: (617) 914-5804

email: [yuxin.jiang@biogen.com](mailto:yuxin.jiang@biogen.com)

## Appendix

<Option1>

```
/ xaxisopts=( Label="Day" offsetmin=0.02 LabelAttrs=( Weight=bold) TickValueAttrs=(Size=9) type=linear linearopts=(  
tickvaluelist=( 0 1 3 6 9 12 15 18 21 24) viewmin=0 viewmax=24 TickValueFitPolicy=Rotate ) )  
y2axisopts=(labelFitPolicy=Split) yaxisopts=( Label="Ratio of two tests" labelFitPolicy=Split LabelAttrs=(Weight=bold)  
type=linear linearopts=( viewmin=2 viewmax=3.7 MinorTicks=true ) ) y2axisopts=(labelFitPolicy=Split)
```